Preliminaries for the "Nullclines that are not lines" example in section 5.2

```
> with(linalg):
> x1:=2*x*(1-x/2)-x*y;
```

$$x1 := 2\,x\left(1-\frac{1}{2}\,x\right)-x\,y$$

```
> y1:=y*(9/4-y^2)-x^2*y;
```

$$y1 := y\left(\frac{9}{4}-y^2\right)-x^2\,y$$

```
> stapt:=solve({x1,y1},{x,y});
```

$$stapt := \{x=0, y=0\}, \{y=0, x=2\}, \{y=\frac{3}{2}, x=0\}, \{y=\frac{-3}{2}, x=0\},$$
$$\{x=2-\frac{1}{2}\,\mathrm{RootOf}(7+2\,\_Z^2-8\,\_Z, label=\_L2), y=\frac{1}{2}\,\mathrm{RootOf}(7+2\,\_Z^2-8\,\_Z, label=\_L2)\}$$

These should be all stationary points. However, the fourth point is on the negative part of the y-axis, while the analysis in the text was confined to the first quadrant. The fifth point is a little troublesome, since it is a single expression that stands for two points.

```
> vecfld:=vector([x1,y1]);
```

$$vecfld := \left[2\,x\left(1-\frac{1}{2}\,x\right)-x\,y, y\left(\frac{9}{4}-y^2\right)-x^2\,y\right]$$

This introduces a single vector expression for the right side of the equation.

Building the Jacobian matrix.

```
> col1:=map(diff,vecfld,x);
```

$$col1 := [2-2\,x-y, -2\,x\,y]$$

```
> col2:=map(diff,vecfld,y);
```

$$col2 := \left[-x, \frac{9}{4}-3\,y^2-x^2\right]$$

```
> J:=augment(col1,col2);
```

$$J := \begin{bmatrix} 2-2\,x-y & -x \\ -2\,x\,y & \frac{9}{4}-3\,y^2-x^2 \end{bmatrix}$$

The origin.

```
> stapt[1];
```

$$\{x=0, y=0\}$$

```
> subs(stapt[1],op(J));
```

$$\begin{bmatrix} 2 & 0 \\ 0 & \frac{9}{4} \end{bmatrix}$$

```
> eigenvects(%);
```

$$[2, 1, \{[1, 0]\}], \left[\frac{9}{4}, 1, \{[0, 1]\}\right]$$

The eigenvalues are both positive and the eigenvectors point along the axes. The origin is an ordinary source.

On the positive x-axis.

```
> stapt[2];
```

$$\{y = 0, x = 2\}$$

```
> subs(stapt[2],op(J));
```

$$\begin{bmatrix} -2 & -2 \\ 0 & \frac{-7}{4} \end{bmatrix}$$

```
> eigenvects(%);
```

$$[-2, 1, \{[1, 0]\}], \left[\frac{-7}{4}, 1, \{[-8, 1]\}\right]$$

Two negative eigenvalues. The point is a sink.

On the positive y-axis.

```
> stapt[3];
```

$$\{y = \frac{3}{2}, x = 0\}$$

```
> subs(stapt[3],op(J));
```

$$\begin{bmatrix} \frac{1}{2} & 0 \\ 0 & \frac{-9}{2} \end{bmatrix}$$

```
> eigenvects(%);
```

$$\left[\frac{1}{2}, 1, \{[1, 0]\}\right], \left[\frac{-9}{2}, 1, \{[0, 1]\}\right]$$

Real eigenvalues of opposite sign. The point is a saddle.

On the negative y-axis.

```
> stapt[4];
```

$$\{y = \frac{-3}{2}, x = 0\}$$

```
> subs(stapt[4],op(J));
```

$$\begin{bmatrix} \frac{7}{2} & 0 \\ 0 & \frac{-9}{2} \end{bmatrix}$$

```
> eigenvects(%);
```

$$\left[\frac{7}{2}, 1, \{[1, 0]\}\right], \left[\frac{-9}{2}, 1, \{[0, 1]\}\right]$$

Another saddle point.

On the circle.

```
> stapt[5];
```

$$\{x = 2 - \frac{1}{2}\text{RootOf}(7 + 2\_Z^2 - 8\_Z, label = \_L2), y = \frac{1}{2}\text{RootOf}(7 + 2\_Z^2 - 8\_Z, label = \_L2)\}$$

There are two points represented by this formula. We need to separate them.

```
> stapt2:=allvalues(stapt[5]);
```

$$stapt2 := \{x = 1 - \frac{1}{4}\sqrt{2}, y = 1 + \frac{1}{4}\sqrt{2}\}, \{y = 1 - \frac{1}{4}\sqrt{2}, x = 1 + \frac{1}{4}\sqrt{2}\}$$

The different roots have conjugate expressions involving the square root of 2.

One point

```
> subs(stapt2[1],op(J));
```

$$\begin{bmatrix} -1 + \frac{1}{4}\sqrt{2} & -1 + \frac{1}{4}\sqrt{2} \\ -2\left(1 - \frac{1}{4}\sqrt{2}\right)\left(1 + \frac{1}{4}\sqrt{2}\right) & \frac{9}{4} - 3\left(1 + \frac{1}{4}\sqrt{2}\right)^2 - \left(1 - \frac{1}{4}\sqrt{2}\right)^2 \end{bmatrix}$$

```
> eigenvects(%);
```

$$\left[-\frac{13}{8} - \frac{3}{8}\sqrt{2} + \frac{1}{8}\sqrt{187 + 22\sqrt{2}}, 1, \left\{\left[-\frac{5}{14} - \frac{5}{14}\sqrt{2} - \frac{1}{14}\sqrt{187 + 22\sqrt{2}}, 1\right]\right\}\right],$$
$$\left[-\frac{13}{8} - \frac{3}{8}\sqrt{2} - \frac{1}{8}\sqrt{187 + 22\sqrt{2}}, 1, \left\{\left[-\frac{5}{14} - \frac{5}{14}\sqrt{2} + \frac{1}{14}\sqrt{187 + 22\sqrt{2}}, 1\right]\right\}\right]$$

Exact expression, but not very useful.

```
> evalf(%);
```

[-.309250208, 1., {[-1.917121916, 1.]}], [-4.001409964, 1., {[.1926836578, 1.]}]

```
> evalf(stapt2[1]);
```

$$\{x = .6464466095, y = 1.353553390\}$$

This shows that the leftmost of the points on the circle is a sink.

The other point.

```
> subs(stapt2[2],op(J));
```

$$\begin{bmatrix} -1-\frac{1}{4}\sqrt{2} & -1-\frac{1}{4}\sqrt{2} \\ -2\left(1-\frac{1}{4}\sqrt{2}\right)\left(1+\frac{1}{4}\sqrt{2}\right) & \frac{9}{4}-3\left(1-\frac{1}{4}\sqrt{2}\right)^2-\left(1+\frac{1}{4}\sqrt{2}\right)^2 \end{bmatrix}$$

```
> eigenvects(%);
```

$$\left[-\frac{13}{8}+\frac{3}{8}\sqrt{2}+\frac{1}{8}\sqrt{187-22\sqrt{2}}, 1, \left\{\left[-\frac{5}{14}+\frac{5}{14}\sqrt{2}-\frac{1}{14}\sqrt{187-22\sqrt{2}}, 1\right]\right\}\right],$$
$$\left[-\frac{13}{8}+\frac{3}{8}\sqrt{2}-\frac{1}{8}\sqrt{187-22\sqrt{2}}, 1, \left\{\left[-\frac{5}{14}+\frac{5}{14}\sqrt{2}+\frac{1}{14}\sqrt{187-22\sqrt{2}}, 1\right]\right\}\right]$$

If you look closely, you will see that the sign of the square root of 2 has changed in all formulas.

```
> evalf(%);
```

$$[.466015541, 1., \{[-.7438868450, 1.]\}], [-2.655355369, 1., \{[1.039753675, 1.]\}]$$

```
> evalf(stapt2[2]);
```

$$\{x = 1.353553390, y = .6464466095\}$$

This shows that the rightmost point on the circle is a saddle point.

Here is a uniform expression for the Jacobian matrix at the two points.

```
> simplify(subs(stapt[5],op(J)));
```

$$\begin{bmatrix} -2+\frac{1}{2}\mathrm{RootOf}(7+2\_Z^2-8\_Z, label=\_L2), & -2+\frac{1}{2}\mathrm{RootOf}(7+2\_Z^2-8\_Z, label=\_L2) \\ \frac{-7}{4}, & \frac{7}{4}-2\,\mathrm{RootOf}(7+2\_Z^2-8\_Z, label=\_L2) \end{bmatrix}$$

Regions in the first quadrant.

Region A. A triangular region along the positive y-axis

Arrows on the boundary of the region point down and to the right. If you have a computer, another approach is to find an interior point and evaluate the diection field there.

```
> subs({x=3/2,y=1/4},op(vecfld));
```

$$\left[\frac{3}{8}, \frac{-1}{64}\right]$$

Note the first entry is positive (to the right) and the second is negative (down). The left side of the region is a trajectory. On the other two sides, the arrows point into the region. The point on the positive y-axis was a saddle: solutions approach the point along the axis, but are repelled horizonally. The point at the right end of the base is the leftmost point on the circle, which we saw to be a sink. There would be trouble if this were not the case, since we have found many ways into the region, but no way out. This poit is the way out. There is one more feature to describe. Some points are on paths entering along the line at the top; others ar on paths entering along the circle on the bottom. The boundary between these types is a separatrix connecting the two stationary vertices of the region.

Region B. The small sector of the circle.

An interior point

```
> subs({x=21/20,y=21/20},op(vecfld));
```

$$\left[\frac{-21}{200}, \frac{189}{4000}\right]$$

Up and to the left. Exactly the opposite of what we saw on the opposite side of the common vertex with A. This is exactly what is expected at a point with a proper linearization. Again, the arrows on the boundary curves point into the region, and the sink at the left corner is the only way out. There is a separatrix from the other stationary vertex (which is a saddle point) dividing the region according to where the trajectory enters the region.

Region C. A triangle along the positive x-axis.

An interior point

```
> subs({x=1/4,y=3/2},op(vecfld));
```

$$\left[\frac{1}{16}, \frac{-3}{32}\right]$$

Down and to the right. Another reflection in a stationary point on the circle. Solutions enter the region along the sides and exit at the sink on the x-axis. A seoaratrix joins the two stationary vertices, as before.

Region D. The bulk of the circle.An interior point

An interior point

```
> subs({x=3/4,y=3/4},op(vecfld));
```

$$\left[\frac{3}{8}, \frac{27}{32}\right]$$

Up and to the right. This was already obvious from the behavior on the axes. Except for the axes, where trajectories run along the side, trajectories exit the region along edges. The only entrance is the source at the origin.There must also be separatrices dividing the region according to the exit side. There are two: one terminating at each of the statinar points on the circle. These paths can only start at the origin.

Region E. The outside.

An interior point

```
> subs({x=3/2,y=3/2},op(vecfld));
```

$$\left[\frac{-3}{2}, \frac{-27}{8}\right]$$

Down and to the left. Towards the circle. Again there will be two separatrices. The techniques for dealing with behavior "near infinity" has not been treated in this course. A general rule for algebraic expressions is to look at the highest degree terms, which are degree 3 in this case and only present in the expression for dy/dt. The other derivative should be treated as zero, so the direction field at distant points is essentially straight down. The separatrices will then originate at the top of any window.